
Ziele, Themen

♦ Pro und Kontra

♦ Grundlegendes im Eilverfahren

♦ Schritt für Schritt-Implementierungen

♦ Beispiele für den Einsatz von Klassen

♦ Anregungen für eigene Klassen

♦ Handreichungen für die Praxis

♦ Lust auf Entwicklung mit Klassen machen

Aufbau

♦ I. Grundlagen
Klassen in VB(A)

♦ II. Mehr als die Summe der Teile
Compound-Controls

♦ III. Datenzugriff elegant
Datenkapselklassen

♦ IV. Angewandtes Wissen
Anwendungsklassen

I. Grundlagen:
Klassen in VB(A)

Klassen und Objekte

Eigenschaften, Methoden und Ereignisse

Benutzerdefinierte Klassen erstellen

Prozedur B

Prozedur C
Daten 2

Objekt2

Prozedur D

Daten 4

Daten 5

Daten 6

Objekt 3

♦ Vereinigung von Daten und dazugehörenden Aktionen

Klassen und Objekte

Prozedur ADaten 1

Daten 3 Prozedur B

Objekt 1

Prozedur B

Prozedur C
Daten 7

Objekt 4

3 Klassen
4 Objekte

2 Instanzen
der selben Klasse

→ „Objekt“

→ „Klasse“

Klassen und Objekte (cont‘d)

♦ Elemente eines Objekts:
• Eigenschaften (=Daten)
• Methoden (=Aktionen)
• Ereignisse

Methoden

Objekt

Ereignisse

Eigenschaften

Lebensdauer eines Objekts

♦ Objekt (Klasseninstanz) erstellen: New
Set frmKundenA = New Form_frmKunden

• mehrfach möglich
Set frmKundenB = New Form_frmKunden

♦ Objekt löschen
• Nicht direkt möglich
• Letzter Verweis gelöscht: Objekt wird gelöscht

♦ Objekt zählt Verweise auf sich

♦ Verweis löschen
Set frmKundenA = Nothing

Zugriff auf Objekt-Elemente

♦ Eigenschaften
<Objekt>.<Eigenschaft>
txtNachname.FontName = "Tahoma"

♦ Methoden
<Objekt>.<Methode>
lstMeals.AddItem "Wiener Schnitzel"

♦ Ereignisse abfangen mit Prozedur
<Objekt>_<Ereignis>([<Parameterliste>])
Private Sub cmdClose_Click()

Eigenschaften erstellen

♦ Öffentliche Modulvariable
Public Nachname As String

♦ Zugriffs-Prozedur/Funktion:
Private mp_strNachname As String

...

Public Property Let Nachname(v As String)

mp_strNachname = v

End Property

Public Property Get Nachname() As String

Nachname = mp_strNachname

End Property

♦ Bei Referenztypen:
Property Set statt Property Let

Methoden erstellen

♦ Methoden sind Prozeduren/Funktionen der Klasse
• In Klassenmodul CKunde:

Public Sub PrintOut()

...

End Sub

• In anderem Modul:
Dim objKunde As CKunde

Set objKunde = New CKunde

...

objKunde.PrintOut()

...

Set objKunde = Nothing

Entscheidungshilfe:
Eigenschaft oder Methode?

♦ Beispiel: Wert „Nachname“ eines Objekts
• Als Eigenschaft?

Public Property Get Nachname() As String

• Als Methode?
Public Function Nachname() As String

• Unterschied:
Vor allem in der Semantik!

♦ Entscheidungshilfe
• Eigenschaft für Zugriff auf Zustände, ohne den Zustand des

Objekts zu verändern (außer Property Let/Set)
• Methode für das Auslösen von Aktionen im Objekt, die i.A.

den Zustand des Objekts verändern

Eigenschaft verwenden!

Wird verworfen
oder gespeichert?♦ Negativ-Beispiel

MyForm.Dirty = False

Ereignisse erstellen

♦ Deklaration
Public Event Changed(OldPrice As Currency)

♦ Auslösen
RaiseEvent Changed(strOldPrice)

♦ Abfangen mit WithEvents (nur in Klassenmodul möglich)
Dim WithEvents meineArtikel As CArtikel

...

Private Sub meinArtikel_Changed( _

OldPrice As Currency)

♦ Rückmeldung an Objekt möglich
• Parameterübergabe ByRef

• Beispiel: Parameter Cancel in Open-Ereignis (Access.Form)

Konstruktion und Destruktion

♦ Klassenereignisse Initialize und Termine
• Ereignisprozeduren

Class_Initialize und Class_Terminate

• Treten bei Beginn bzw. Ende der Lebensdauer auf
• Für Initialisierungen bzw. Freigabe von Ressourcen

♦ Class_Initialize: Keine Parameterübergabe bei
Konstruktion möglich
• Gefahr unvollständig initialisierter Objekte
• Abhilfe: Factory-Funktion

- Beispiel (DAO): OpenRecordset

Vergleich Standardmodul vs. Klassenmodul

Qualifiziert mit
Objektname

Objekt.Test()

Direkt
Test()

oder

qualifiziert mit
Modulname

Modul.Test()

Öffentliche
Eigenschaften/Methoden
ansprechen

JaNeinAls Ereignissenke
JaNeinAls Ereignisquelle

Beliebig vieleGenau 1Instanzen

Explizit
Set Objekt = New Klasse

KeineInstanziierung
KlassenmodulStandardmodul

Szenario

♦ Beteiligte Personen und Begriffe

KlassenAnwendung

Anwender „DevUser“ Entwickler
wendet an

besteht aus

en
tw

ick
elt

en
tw

ick
elt verwendet

II. Mehr als die Summe der Teile
Compound-Controls

Idee

Implementierung

Tipps und Tricks

Motivation

♦ Idee
• Neue Funktionalitäten durch Kombination vorhandener

Steuerelemente

♦ Wunsch
• Möglichst einfache Handhabung
• Ähnlich nativer Steuerelemente

♦ Beispiel
• ProgressBar

- Rechteck
- Bezeichnungsfeld

Motivation (cont‘d)

♦ Weitere Beispiele
• CoolButton

- Rechteck
- Bezeichnungsfeld
- Transparente Schaltfläche

• Sprengel
- Mehrere Bezeichnungsfelder
- Bild

• MoveAndSizeLabel
- Bezeichnungsfeld

Implementierungsbeispiel:
CcProgressBar

♦ Grafische Darstellung eines Fortschritts:

Rechteck
Bezeichnungsfeld

♦ CcProgressBar soll folgende Features bieten
• Wert zwischen 0 und 1
• Position beliebig
• Größe beliebig
• Abstand Rahmen-Balken einstellbar

Implementierung CcProgressBar:
Eigenschaften

Breite des SteuerelementsLongWidth

Abstand zwischen äußerem Rand und BalkenLongPadding

Anzuzeigender Wert (0..1)DoubleValue

Höhe des SteuerelementsLongHeight

Position der oberen Kante des SteuerelementsLongTop

Position der linken Kante des SteuerelementsLongLeft

Der BalkenAccess.LabelBarLabel

Äußerer Rahmen des SteuerelementsAccess.RectangleFrameRectangle

BeschreibungDatentypEigenschaft

♦ Folgende Eigenschaften sind notwendig:
Bestimmt Position

und Größe

Implementierung CcProgressBar:
Speicherort der Eigenschaftswerte

mp_rctFrame.WidthWidth

mp_rctFrame.Height

mp_rctFrame.Top

mp_lngPadding

mp_dblValue

mp_lblBar

mp_rctFrame.Left

mp_rctFrame

Speicherort

Steuerelement-
Eigenschaft

Member-Variable

Art

Value

Padding

Height

Top

Left

BarLabel

FrameRectangle

Eigenschaft

♦ Die Eigenschaften werden wie folgt gespeichert:

Implementierung CcProgressBar:
Interne Berechnungen

♦ Notwendige interne Berechnungen (SetupBarLabel)
mp_lblBar.Left = mp_rctFrame.Left +    mp_lngPadding

mp_lblBar.Width = mp_dblValue *
(mp_rctFrame.Width - 2 * mp_lngPadding)

mp_lblBar.Top = mp_rctFrame.Top +    mp_lngPadding

mp_lblBar.Height = mp_rctFrame.Height - 2 * mp_lngPadding

♦ Abhängigkeiten von öffentlichen Eigenschaften:

mp_rctFrame.Left = Left

mp_rctFrame.Top = Top

mp_rctFrame.Width = Width

mp_rctFrame.Height = Height

mp_rctFrame = FrameRectangle

mp_lblBar = BarLabel

mp_dblValue = Value

mp_lngPadding = Padding

Implementierung CcProgressBar:
Abhängigkeiten

♦ Es existieren folgende Abhängigkeiten:

mp_lblBar.Left

mp_lngPadding

mp_rctFrame.Left

Left TopWidth Height

mp_dblValue

mp_rctFrame.Topmp_rctFrame.Width

mp_lblBar.Topmp_lblBar.Width mp_lblBar.Height

PaddingValue

mp_rctFrame.Height

Frame-
Rectangle

BarLabel

mp_rctFrame mp_lblBar

Implementierung CcProgressBar:
Verwendung

♦ Verwendung in einem Formular:
Private ProgressBar As CcProgressBar

Private Sub Form_Open(Cancel As Integer)

Set ProgressBar = New CcProgressBar

With ProgressBar

Set .BarLabel = Me.lblProgress

Set .FrameRectangle = Me.rctProgress

End With

End Sub

Private Sub Form_Close()

Set ProgressBar = Nothing

End Sub

...

ProgressBar.Value = 0.5

...

Steuerelement-
Setup

Aufräumen

Verwendung



CcProgressBarSimple

' This is CcProgressBarSimple
Option Compare Database
Option Explicit

Private mp_rctFrame As Access.Rectangle
Private mp_lblBar As Access.Label
Private mp_lngPadding As Long
Private mp_dblValue As Double

' ___ Initialize/Terminate ___

Private Sub Class_Initialize()
mp_lngPadding = 60
mp_dblValue = 0#

End Sub

' ___ Property Procedures ___

Public Property Set FrameRectangle(ByRef NewFrameRectangle As Access.Rectangle)
Set mp_rctFrame = NewFrameRectangle
SetupBarControl

End Property
Public Property Get FrameRectangle() As Access.Rectangle

Set FrameRectangle = mp_rctFrame
End Property

Public Property Set BarLabel(ByRef NewBarLabel As Access.Label)
Set mp_lblBar = NewBarLabel
SetupBarControl

End Property
Public Property Get BarLabel() As Access.Label

Set BarLabel = mp_lblBar
End Property

Public Property Let Left(ByVal NewLeft As Long)
mp_rctFrame.Left = NewLeft
SetupBarControl

End Property
Public Property Get Left() As Long

Left = mp_rctFrame.Left
End Property

Public Property Let Top(ByVal NewTop As Long)
mp_rctFrame.Top = NewTop
SetupBarControl

End Property
Public Property Get Top() As Long

Top = mp_rctFrame.Top
End Property

Public Property Let Width(ByVal NewWidth As Long)
mp_rctFrame.Width = NewWidth
SetupBarControl

End Property
Public Property Get Width() As Long

Width = mp_rctFrame.Width
End Property

Public Property Let Height(ByVal NewHeight As Long)
mp_rctFrame.Height = NewHeight
SetupBarControl

End Property
Public Property Get Height() As Long

Height = mp_rctFrame.Height
End Property

Public Property Let Padding(ByVal NewPadding As Long)
mp_lngPadding = NewPadding
SetupBarControl

End Property
Public Property Get Padding() As Long

Padding = mp_lngPadding
End Property



CcProgressBarSimple

Public Property Let Value(ByVal NewValue As Double)
mp_dblValue = NewValue

With mp_lblBar
.Width = mp_dblValue * (mp_rctFrame.Width - 2 * mp_lngPadding)
.Caption = Format(mp_dblValue, "Percent")

End With
End Property
Public Property Get Value() As Double

Value = mp_dblValue
End Property

' ___ Private Methods ___

Private Sub SetupBarControl()
If mp_lblBar Is Nothing Then Exit Sub
If mp_rctFrame Is Nothing Then Exit Sub

With mp_lblBar
.Left = mp_rctFrame.Left + mp_lngPadding
.Top = mp_rctFrame.Top + mp_lngPadding
.Width = mp_dblValue * (mp_rctFrame.Width - 2 * mp_lngPadding)
.Height = mp_rctFrame.Height - 2 * mp_lngPadding

End With
End Sub

Implementierung CcProgressBar:
Robustheit

♦ Probleme mit der aktuellen Implementierung
• Laufzeitfehler bei negativen Werten für

- Left
- Top
- Width
- Height

• Eigenschaft Value:
- Negative Werte und Werte > 1:

Inkonsistente Darstellung
- Negative Werte (<<):

Laufzeitfehler

♦ Es fehlt: Definition der Randbedingungen

Ev. Wahl zwischen diesen Varianten:
- Eigenschaft („SilentMode“)
- Bedingte Kompilierung

Ev. Wahl zwischen diesen Varianten:
- Eigenschaft („SilentMode“)
- Bedingte Kompilierung

Implementierung CcProgressBar:
Unzulässige Eingangswerte

♦ Was tun bei unzulässigen Eingangswerten?

♦ 3 Möglichkeiten
• Laufzeitfehler auslösen

- Wenn damit das Cc nicht mehr funktionsfähig werden würde
If NewLabel Is Nothing Then Err.Raise ...

• Unzulässigen Wert korrigieren
- Wenn der Wert „anstoßen“ darf

If NewWidth < 0 Then NewWidth = 0

- Nicht bei Datenwerten (z.B. Preis)

• Unzulässigen Wert ignorieren
If NewValue < 0 Then Exit Property

Implementierung CcProgressBar:
Randbedingungen

♦ Randbedingungen für öffentliche Eigenschaften:

0 setzen≥ 0Width

0 setzen≥ 0Padding

0 bzw. 1 setzen≥ 0Value

0 setzen≥ 0Height

0 setzen≥ 0Top

0 setzen≥ 0Left

LaufzeitfehlerNot NothingBarLabel

LaufzeitfehlerNot NothingFrameRectangle

Reaktion bei VerletzungRandbedingungEigenschaft

Implementierung CcProgressBar:
Randbedingungen (cont‘d)

♦ Randbedingungen für interne Eigenschaften:

0 setzen≥ 0mp_lblBar.Width

0 setzen≥ 0mp_rctFrame.Height

0 setzen≥ 0mp_rctFrame.Top

0 setzen≥ 0mp_lblBar.Height

0 setzen≥ 0mp_lblBar.Top

0 setzen≥ 0mp_lblBar.Left

0 setzen≥ 0mp_rctFrame.Width

0 setzen≥ 0mp_rctFrame.Left

Reaktion bei VerletzungRandbedingungEigenschaft

Paul Rohorzka Klassen konkret 30

Einschub:
Beschränkte Zuweisung von Werten

♦ Aufgabenstellung: Wert muss über einer Grenze sein

♦ Fall 1: Zuweisung des falschen Wertes möglich
MyValue = NewValue // Ev. unter der Grenze

If MyValue < MinValue Then MyValue = MinValue

♦ Fall 2: Wert darf nie außerhalb der Grenze sein
If NewValue >= MinValue Then

MyObject.Prop = NewValue

Else

MyObject.Prop = MinValue

End If

• Relativ viel Code
• Doppelte Berechnungen, wenn NewValue ein Ausdruck ist

Einschub:
Beschränkte Zuweisung von Werten (cont‘d)

♦ Vorschlag: Verwendung einer Methode Max():
Private Function Max(a As Long, b As Long) As Long

Max = Iif(a > b, a, b)

End Function

♦ Anwendung:
MyObject.Prop = Max(MinValue, NewValue)

• Weniger Code
• Auswertung von NewValue nur einmal
• Performance? (Funktionsaufrufe)

♦ Beschränkung auf Ober- und Untergrenze:
MyObject.Prop = Max(MinValue, _

Min(MaxValue, NewValue) _
)

Implementierung CcProgressBar:
Durchsetzung der Randbedingungen

♦ Eigenschafts-Prozedur Form 1:
Public Property Let Left(ByVal NewLeft As Long)

If (mp_rctFrame Is Nothing) Then

Err.Raise ...

End If

mp_rctFrame.Left = Max(0, NewLeft)

SetupBarControl

End Property

Public Property Get Left() As Long

Left = mp_rctFrame.Left

End Property

Implementierung CcProgressBar:
Durchsetzung der Randbedingungen (cont‘d)

♦ Eigenschafts-Prozedur Form 2:
Public Property Let Value(ByVal NewValue As Double)

mp_dblValue = NewValue

If mp_dblValue < 0 Then mp_dblValue = 0

If mp_dblValue > 1 Then mp_dblValue = 1

If (mp_rctFrame Is Nothing) Then Exit Property

If (mp_lblBar Is Nothing) Then Exit Property

With mp_lblBar

.Width = mp_dblValue * _
Max(0, mp_rctFrame.Width _

- 2 * mp_lngPadding)

.Caption = Format(mp_dblValue, "Percent")

End With

End Property



CcProgressBarBetter

' This is CcProgressBarBetter
Option Compare Database
Option Explicit

Private mp_rctFrame As Access.Rectangle
Private mp_lblBar As Access.Label
Private mp_lngPadding As Long
Private mp_dblValue As Double

' ___ Initialize/Terminate ___

Private Sub Class_Initialize()
mp_lngPadding = 60
mp_dblValue = 0#

End Sub

' ___ Property Procedures ___

Public Property Set FrameRectangle(ByRef NewFrameRectangle As Access.Rectangle)
If (NewFrameRectangle Is Nothing) Then

Err.Raise vbObjectError + 1000, TypeName(Me), "Die Eigenschaft FrameRectangle darf nicht Nothing
sein."

End If
Set mp_rctFrame = NewFrameRectangle
SetupBarControl

End Property
Public Property Get FrameRectangle() As Access.Rectangle

Set FrameRectangle = mp_rctFrame
End Property

Public Property Set BarLabel(ByRef NewBarLabel As Access.Label)
If (NewBarLabel Is Nothing) Then

Err.Raise vbObjectError + 1001, TypeName(Me), "Die Eigenschaft BarLabel darf nicht Nothing sein."
End If
Set mp_lblBar = NewBarLabel
SetupBarControl

End Property
Public Property Get BarLabel() As Access.Label

Set BarLabel = mp_lblBar
End Property

Public Property Let Left(ByVal NewLeft As Long)
If (mp_rctFrame Is Nothing) Then

Err.Raise vbObjectError + 1002, TypeName(Me), "Das Objekt ist nicht korrekt initialisiert. Die
erforderliche Eigenschaft FrameRectangle ist nicht gesetzt."

End If
mp_rctFrame.Left = Max(0, NewLeft)
SetupBarControl

End Property
Public Property Get Left() As Long

Left = mp_rctFrame.Left
End Property

Public Property Let Top(ByVal NewTop As Long)
If (mp_rctFrame Is Nothing) Then

Err.Raise vbObjectError + 1002, TypeName(Me), "Das Objekt ist nicht korrekt initialisiert. Die
erforderliche Eigenschaft FrameRectangle ist nicht gesetzt."

End If
mp_rctFrame.Top = Max(0, NewTop)
SetupBarControl

End Property
Public Property Get Top() As Long

Top = mp_rctFrame.Top
End Property

Public Property Let Width(ByVal NewWidth As Long)
If (mp_rctFrame Is Nothing) Then

Err.Raise vbObjectError + 1002, TypeName(Me), "Das Objekt ist nicht korrekt initialisiert. Die
erforderliche Eigenschaft FrameRectangle ist nicht gesetzt."

End If
mp_rctFrame.Width = Max(0, NewWidth)
SetupBarControl

End Property



CcProgressBarBetter

Public Property Get Width() As Long
Width = mp_rctFrame.Width

End Property

Public Property Let Height(ByVal NewHeight As Long)
If (mp_rctFrame Is Nothing) Then

Err.Raise vbObjectError + 1002, TypeName(Me), "Das Objekt ist nicht korrekt initialisiert. Die
erforderliche Eigenschaft FrameRectangle ist nicht gesetzt."

End If
mp_rctFrame.Height = Max(0, NewHeight)
SetupBarControl

End Property
Public Property Get Height() As Long

Height = mp_rctFrame.Height
End Property

Public Property Let Padding(ByVal NewPadding As Long)
mp_lngPadding = Max(0, NewPadding)
SetupBarControl

End Property
Public Property Get Padding() As Long

Padding = mp_lngPadding
End Property

Public Property Let Value(ByVal NewValue As Double)
mp_dblValue = NewValue
If mp_dblValue < 0 Then mp_dblValue = 0
If mp_dblValue > 1 Then mp_dblValue = 1
If (mp_rctFrame Is Nothing) Then Exit Property
If (mp_lblBar Is Nothing) Then Exit Property

With mp_lblBar
.Width = mp_dblValue * Max(0, (mp_rctFrame.Width - 2 * mp_lngPadding))
.Caption = Format(mp_dblValue, "Percent")

End With
End Property
Public Property Get Value() As Double

Value = mp_dblValue
End Property

' ___ Private Methods ___

Private Sub SetupBarControl()
If mp_lblBar Is Nothing Then Exit Sub
If mp_rctFrame Is Nothing Then Exit Sub

With mp_lblBar
.Left = mp_rctFrame.Left + IIf(mp_rctFrame.Width > mp_lngPadding, mp_lngPadding, mp_rctFrame.Width)
.Width = mp_dblValue * Max(0, mp_rctFrame.Width - 2 * mp_lngPadding)
.Height = Max(0, mp_rctFrame.Height - 2 * mp_lngPadding)
.Top = mp_rctFrame.Top + IIf(mp_rctFrame.Height > mp_lngPadding, mp_lngPadding, mp_rctFrame.Height)

End With
End Sub

Private Function Min(ByVal a As Long, ByVal b As Long) As Long
Min = IIf(a < b, a, b)

End Function

Private Function Max(ByVal a As Long, ByVal b As Long) As Long
Max = IIf(a > b, a, b)

End Function

CcProgressBar:
Erweiterung

♦ Objekt soll bei Erreichen eines Schwellwertes ein
Ereignis auslösen

♦ Zusätzliche Eigenschaft für den Schwellwert

Wert, bei dessen Überschreiten das Ereignis Alarm
ausgelöst werden soll.

DoubleAlarmLevel
BeschreibungDatentypEigenschaft

♦ Neu: Alarm-Ereignis mit Information über Richtung
Public Event Alarm(ByVal IncreasingValue _

As Boolean)

♦ Interne Variablen:
m_blnIncreasingValue

m_blnAlarmed

m_dblOldValue

Implementierung CProgressBar:
Alarm-Ereignis

♦ Implementierung des Alarms
Public Property Let Value(ByVal NewValue As Double)

m_dblOldValue = mp_dblValue

mp_dblValue = NewValue

...

If mp_dblValue > m_dblOldValue Then

If mp_dblValue >= mp_dblAlarmLevel _
And Not m_blnAlarmed Then

RaiseEvent Alarm(True)

m_blnAlarmed = True

End If

If Not m_blnIncreasingValue Then

m_blnIncreasingValue = True

m_blnAlarmed = False

End If

ElseIf mp_dblValue < m_dblOldValue Then

... ' entsprechend für kleiner werdenden Wert

End If

End Property

Auslösen des
Alarms

Implementierung CProgressBar:
Alarm-Ereignis (cont‘d)

♦ Reaktion auf den Alarm im Formular:
• Auf Modulebene:

Ergänzung der Deklaration mit WithEvents
Private WithEvents ProgressBar As CcProgressBarAlarm

• Implementierung der Ereignisprozedur
(mit Editorunterstützung):

Private Sub ProgressBar_Alarm(ByVal IncreasingValue _
As Boolean)

MsgBox(...)

End Sub



CcProgressBarAlarm (Auszug)

' This is CcProgressBarAlarm
Option Compare Database
Option Explicit

Public Event Alarm(ByVal IncreasingValue As Boolean)

Private mp_rctFrame As Access.Rectangle
Private mp_lblBar As Access.Label
Private mp_lngPadding As Long
Private mp_dblValue As Double
Private mp_dblAlarmLevel As Double

Private m_blnIncreasingValue As Boolean
Private m_blnAlarmed As Boolean
Private m_dblOldValue As Double

' ___ Initialize/Terminate ___

Private Sub Class_Initialize()
mp_lngPadding = 60
mp_dblValue = 0#
mp_dblAlarmLevel = mp_dblValue

m_blnIncreasingValue = True
m_dblOldValue = mp_dblValue
m_blnAlarmed = False

End Sub

' ___ Property Procedures ___

...

Public Property Let Value(ByVal NewValue As Double)
m_dblOldValue = mp_dblValue
mp_dblValue = NewValue
If mp_dblValue < 0 Then mp_dblValue = 0
If mp_dblValue > 1 Then mp_dblValue = 1
If (mp_rctFrame Is Nothing) Then Exit Property
If (mp_lblBar Is Nothing) Then Exit Property

With mp_lblBar
.Width = mp_dblValue * Max(0, (mp_rctFrame.Width - 2 * mp_lngPadding))
.Caption = Format(mp_dblValue, "Percent")

End With

If mp_dblValue > m_dblOldValue Then
If mp_dblValue >= mp_dblAlarmLevel And Not m_blnAlarmed Then

RaiseEvent Alarm(True)
m_blnAlarmed = True

End If
If Not m_blnIncreasingValue Then

m_blnIncreasingValue = True
m_blnAlarmed = False

End If
ElseIf mp_dblValue < m_dblOldValue Then

If mp_dblValue <= mp_dblAlarmLevel And Not m_blnAlarmed Then
RaiseEvent Alarm(False)
m_blnAlarmed = True

End If
If m_blnIncreasingValue Then

m_blnIncreasingValue = False
m_blnAlarmed = False

End If
End If

End Property
Public Property Get Value() As Double

Value = mp_dblValue
End Property

Public Property Let AlarmLevel(ByVal NewAlarmLevel As Double)
mp_dblAlarmLevel = NewAlarmLevel
If mp_dblAlarmLevel < 0 Then mp_dblAlarmLevel = 0
If mp_dblAlarmLevel > 1 Then mp_dblAlarmLevel = 1

End Property
Public Property Get AlarmLevel() As Double

AlarmLevel = mp_dblAlarmLevel
End Property

...

CcProgressBar:
Weitere Features

♦ Farben
• BackColor, BarColor, TextColor

♦ Beliebiger Bereich Value
• Minimum, Maximum

♦ Orientierung (waagrecht – senkrecht)

♦ Format des anzuzeigenden Textes

♦ Interaktion: Einstellen des Wertes mit der Maus

Compound-Controls:
Hinweise zur Entwicklung

♦ Von einem Access-Steuerelement Position und Größe
ableiten
• Kapselung (DevUser-Code unabhängig von Änderungen im Cc)
• Begriffe auf Ebene des Compound-Controls

♦ DevUser:
• Alle Manipulationen über das Compound-Control
• Nicht direkt über Access-Steuerelemente

Compound-Controls:
Grenzen

♦ Nicht Endlosformular-tauglich

♦ Nicht konsequent gekapselt

♦ Beschränkte Anzahl der Steuerelemente

♦ Z-Reihenfolge der Steuerelemente ist ev. relevant,
kann jedoch nicht verändert werden
• Muss in der Entwurfsansicht korrekt sein

(Verantwortung des DevUsers)

Compound-Controls:
Aus Sicht des DevUsers

♦ Sollen wie Access-Steuerelemente verwendbar sein

♦ Was zusätzlich gemacht werden muss:
• Deklaration einer Member-Variablen
• Instanziierung
• Setup

- Hineinreichen von Access-Steuerelementen
- Initialisierung relevanter Eigenschaften

• Deinstanziierung

Paul Rohorzka Klassen konkret 43

Compound-Controls:
Aspekte aus Sicht des Entwicklers

♦ Austausch der Access-Steuerelemente verhindern
• WriteOnce-Eigenschaften

oder
• Laufzeitfehler bei Änderung
• ReadOnly-Eigenschaften und Init-Methode

♦ Möglichst leichtgewichtige Access-Steuerelemente
• Bezeichnungsfelder sind vielseitig,

aber
• Keine Events und kein Text: Rechteck-Steuerelement

♦ Maximale Robustheit
• Mit unzulässige Werte rechnen!

Paul Rohorzka Klassen konkret 44

Compound-Controls:
Aspekte aus Sicht des Entwicklers (cont‘d)

♦ Nach dem Setup möglichst keine Laufzeitfehler
auslösen
• Unzulässige Werte ignorieren oder korrigieren

♦ Vermeidung unnötigen Neuzeichnens
• Aktion nur bei veränderten Werten
• Möglichst nur veränderte Werte erneut zuweisen
• ev. ImmediateUpdate-Eigenschaft implementieren

Prinzipien im Umgang mit Fehlern

♦ Fehlermöglichkeiten für DevUser minimieren!
• Dokumentation der öffentlichen Schnittstelle
• Demo-Programm

- Manipulation aller Eigenschaften
- Aufruf aller Methoden
- Beobachtung der Reaktion

• Aussagekräftige Namen
- Beibehaltung etablierter Begriffe

bspw. Left, nicht x
- Keine Neuinterpretation etablierte Begriffe

nicht Left(), wenn MoveLeft() gemeint ist

• Methode Init zur Festlegung aller erforderlichen Parameter

Paul Rohorzka Klassen konkret 46

Prinzipien im Umgang mit Fehlern (cont‘d)

♦ Möglichst viele Compiler-Fehler!
• Schnittstelle eindeutig gestalten
• „Kleine“ Datentypen verwenden
• Aufzählungstypen verwenden wo sinnvoll

♦ Laufzeitfehler schon beim Setup!
• Notwendige Laufzeitfehler möglichst früh auslösen

(einfacheres Debugging im Setup – Form_Open)

♦ Keine Fehler zur restlichen Laufzeit!
• Wenn möglich: korrigieren oder ignorieren

♦ Konkrete Fehlermeldungen!
• Pfui: „Ungültiger Eigenschaftswert.“
• Besser: „Der Preis darf nicht negativ werden.“

III. Datenzugriff elegant
Datenkapselklassen

Aufgabenstellungen

Implementierungen

Auflistungsklassen

Das Problem:
Hantieren von Daten per Code

♦ Anforderung
• Lesen und/oder schreiben von Daten abseits eines Formulars

♦ Implementierung ist möglich:
intAnzahl = DCount(...)

• besser (?):
Dim rst As DAO.Recordset

Set rst = CurrentDb().OpenRecordset(...)

intAnzahl = rst!Anzahl

rst.Close

Set rst = Nothing
So viel Tipperei
für eine Zahl?

Pfui!

Lösung:
Datenkapselklassen

♦ Abhilfe
• Implementieren von Kapselklassen um die Tabellen

♦ Vorteile
• Objektbasierter Zugriff auf die Daten
• Unterstützung durch Intellisense
• Code liest sich beinahe wie Prosa

♦ Nachteile
• Implementierung der Klassen ist nicht trivial (siehe Folgendes)
• Wer implementiert die Klassen?

Beispiel:
Datenkapselklassen für eine Kunden-Tabelle

♦ Struktur Tabelle tblKunden

erforderlichText (20)Knd_GeaendertVon
erforderlichDatum/UhrzeitKnd_GeaendertAm

erforderlich, >0WährungKnd_Freigrenze

Datum/UhrzeitKnd_Geburtsdatum

erforderlichText (50)Knd_Nachname

Text (30)Knd_Vorname
PKLongKundenID

RegelnDatentypFeld

♦ Objektmodell
(vorläufig)

CKundeCKunde

CKunden

♦ Disconnected

Implementierung CKunde:
Erste Eigenschaften

♦ Member-Variablen für Daten-Eigenschaften
mpd_<ppp><Nnnn> (z.B. mpd_strVorname)

♦ Eigenschaft Dirty (Read only)
mp_blnDirty

♦ Eigenschafts-Prozeduren für Daten-Eigenschaften
Public Property Let Vorname(ByVal NewVorname As String)

NewVorname = Trim$(NewVorname)

mp_blnDirty = mp_blnDirty Or _
(NewVorname <> mpd_strVorname)

mpd_strVorname = NewVorname

End Property

Public Property Get Vorname() As String

Vorname = mpd_strVorname

End Property

♦ Standard-Eigenschaft AsString (Read only)
Public Property Get AsString() As String

Implementierung CKunde:
Lesen und Schreiben der Daten

♦ Konstante für Tabellennamen
Private Const mc_strTableName As String = "tblKunden"

♦ Konstanten für Feldnamen
Private Const mcf_strVorname As String = "Knd_Vorname"
...

♦ Lesen auf zwei Arten:
• Mit übergebener ID über neues Recordset

Public Sub Load(ByVal KundenID As Long)

• Von externem Recordset
Public Sub LoadFromRecordset( _

ByRef rst As DAO.Recordset)

♦ Schreiben auf zwei Arten:
• Über neues Recordset

Public Sub Save()

• Über externes Recordset
Public Sub SaveToRecordset( _

ByRef rst As DAO.Recordset)

Implementierung CKunde:
Lesen und Schreiben der Daten (cont'd)

♦ Lesen von neuem Recordset
Public Sub Load(ByVal KundenID As Long)

Dim strSQL As String

Dim rst As DAO.Recordset

strSQL = "SELECT * " & _

"FROM " & mc_strTableName & " " & _

"WHERE " & mcf_strID & " = " & KundenID

Set rst = CurrentDb().OpenRecordset(strSQL, _
dbOpenDynaset)

LoadFromRecordset rst

If Not (rst Is Nothing) Then rst.Close

Set rst = Nothing

End Sub

Implementierung CKunde:
Lesen und Schreiben der Daten (cont'd)

♦ Lesen von externem Recordset
Public Sub LoadFromRecordset( _

ByRef rst As DAO.Recordset)

With rst

mpd_lngID = .Fields(mcf_strID).Value

mpd_strVorname = _
Nz(.Fields(mcf_strVorname), "")

mpd_strNachname = .Fields(mcf_strNachname)

mpd_dtmGeburtsdatum = _
Nz(.Fields(mcf_strGeburtsdatum), CDate(0))

...

End With

mp_blnDirty = False

End Sub

Implementierung CKunde:
Lesen und Schreiben der Daten (cont'd)

♦ Ähnlich schreiben:
Public Sub SaveToRecordset(ByRef rst As DAO.Recordset)

If Not mp_blnDirty Then Exit Sub

mpd_dtmGeaendertAm = Now()
mpd_strGeaendertVon = CurrentUser()

With rst
.Edit
.Fields(mcf_strVorname) = _

iNz(mpd_strVorname, "")
.Fields(mcf_strNachname) = mpd_strNachname
.Fields(mcf_strGeburtsdatum) = _

iNz(mpd_dtmGeburtsdatum, CDate(0))
...
.Update

End With
mp_blnDirty = False

End Sub

"Inverses Nz()"
Liefert NULL statt
bestimmtem Wert

♦ Test schreiben
Public Sub TestKundeSimpleSave()

Dim Kunde As CKundeSimple

Set Kunde = New CKundeSimple
Kunde.Load 1000
Kunde.Vorname = "Maria"
Kunde.Save
Set Kunde = Nothing

End Sub

Implementierung CKunde:
Erster Test

♦ Test lesen
Public Sub TestKundeSimpleLoad()

Dim Kunde As CKundeSimple

Set Kunde = New CKundeSimple
Kunde.Load 1000
Debug.Print Kunde
Set Kunde = Nothing

End Sub

Verwendung
OverheadOverhead

Verwendung
OverheadOverhead

' This is CKundeSimple
Option Compare Database
Option Explicit

' Constants
Private Const mc_strTableName As String = "tblKunden"

Private Const mcf_strID As String = "KundenID"
Private Const mcf_strVorname As String = "Knd_Vorname"
Private Const mcf_strNachname As String = "Knd_Nachname"
Private Const mcf_strGeburtsdatum As String = "Knd_Geburtsdatum"
Private Const mcf_strFreigrenze As String = "Knd_Freigrenze"
Private Const mcf_strGeaendertAm As String = "Knd_GeaendertAm"
Private Const mcf_strGeaendertVon As String = "Knd_GeaendertVon"

' Member variables holding non-data property values
Private mp_blnDirty As Boolean
Private mp_blnAutoSave As Boolean

' Member variables holding data-property values
Private mpd_lngID As Long
Private mpd_strVorname As String
Private mpd_strNachname As String
Private mpd_dtmGeburtsdatum As Date
Private mpd_curFreigrenze As Currency
Private mpd_dtmGeaendertAm As Date
Private mpd_strGeaendertVon As String

' Member variables for internal use

' ___ Initialize/Terminate ___

Private Sub Class_Initialize()
mp_blnDirty = False
mp_blnAutoSave = True

End Sub

Private Sub Class_Terminate()
If mp_blnAutoSave Then

Save
End If

End Sub

' ___ Public Properties ___

'-- Read only
Public Property Get ID() As Long

ID = mpd_lngID
End Property

Public Property Let Vorname(ByVal NewVorname As String)
NewVorname = Trim$(NewVorname)
mp_blnDirty = mp_blnDirty Or (NewVorname <> mpd_strVorname)
mpd_strVorname = NewVorname

End Property
Public Property Get Vorname() As String

Vorname = mpd_strVorname
End Property

Public Property Let Nachname(ByVal NewNachname As String)
NewNachname = Trim$(NewNachname)
mp_blnDirty = mp_blnDirty Or (NewNachname <> mpd_strNachname)
mpd_strNachname = NewNachname

End Property
Public Property Get Nachname() As String

Nachname = mpd_strNachname
End Property

Public Property Let Geburtsdatum(ByVal NewGeburtsdatum As Date)
mp_blnDirty = mp_blnDirty Or (NewGeburtsdatum <> mpd_dtmGeburtsdatum)
mpd_dtmGeburtsdatum = NewGeburtsdatum

End Property

Public Property Get Geburtsdatum() As Date
Geburtsdatum = mpd_dtmGeburtsdatum

End Property

Public Property Let Freigrenze(ByVal NewFreigrenze As Currency)
mp_blnDirty = mp_blnDirty Or (NewFreigrenze <> mpd_curFreigrenze)
mpd_curFreigrenze = NewFreigrenze

End Property
Public Property Get Freigrenze() As Currency

Freigrenze = mpd_curFreigrenze
End Property

'-- Read only
Public Property Get GeaendertAm() As Date

GeaendertAm = mpd_dtmGeaendertAm
End Property

'-- Read only
Public Property Get GeaendertVon() As String

GeaendertVon = mpd_strGeaendertVon
End Property

'-- Read only
Public Property Get Dirty()

Dirty = mp_blnDirty
End Property

'-- Read only
Public Property Get AsString() As String
'Attribute AsString.VB_UserMemId = 0

AsString = mpd_strNachname
If Len(mpd_strVorname) > 0 Then

AsString = AsString & ", " & mpd_strVorname
End If

End Property

Public Property Let AutoSave(ByVal NewAutoSave As Boolean)
mp_blnAutoSave = NewAutoSave

End Property
Public Property Get AutoSave() As Boolean

AutoSave = mp_blnAutoSave
End Property

' ___ Public Methods ___

Public Sub Load(ByVal KundenID As Long)
Dim strSQL As String
Dim rst As DAO.Recordset

strSQL = "SELECT * " & _
"FROM " & mc_strTableName & " " & _
"WHERE " & mcf_strID & " = " & KundenID

Set rst = CurrentDb().OpenRecordset(strSQL, dbOpenDynaset)
LoadFromRecordset rst
If Not (rst Is Nothing) Then rst.Close
Set rst = Nothing

End Sub

Public Sub LoadFromRecordset(ByRef rst As DAO.Recordset)
With rst

mpd_lngID = .Fields(mcf_strID).Value
mpd_strVorname = Nz(.Fields(mcf_strVorname).Value, "")
mpd_strNachname = .Fields(mcf_strNachname).Value
mpd_dtmGeburtsdatum = Nz(.Fields(mcf_strGeburtsdatum).Value, CDate(0))
mpd_curFreigrenze = .Fields(mcf_strFreigrenze).Value
mpd_dtmGeaendertAm = .Fields(mcf_strGeaendertAm).Value
mpd_strGeaendertVon = .Fields(mcf_strGeaendertVon).Value

End With
mp_blnDirty = False

End Sub

Public Sub Save()
Dim strSQL As String
Dim rst As DAO.Recordset

strSQL = "SELECT * " & _
"FROM " & mc_strTableName & " " & _
"WHERE " & mcf_strID & " = " & mpd_lngID

Set rst = CurrentDb().OpenRecordset(strSQL, dbOpenDynaset)
SaveToRecordset rst
If Not (rst Is Nothing) Then rst.Close
Set rst = Nothing

End Sub

Public Sub SaveToRecordset(ByRef rst As DAO.Recordset)
If Not mp_blnDirty Then Exit Sub

mpd_dtmGeaendertAm = Now()
mpd_strGeaendertVon = CurrentUser()

With rst
If mpd_lngID = 0 Then

.AddNew
mpd_lngID = .Fields(mcf_strID).Value

Else
.Edit

'         .Fields(mcf_strID).Value = mpd_lngID
End If
.Fields(mcf_strVorname).Value = iNz(mpd_strVorname, "")
.Fields(mcf_strNachname).Value = mpd_strNachname
.Fields(mcf_strGeburtsdatum).Value = iNz(mpd_dtmGeburtsdatum, CDate(0))
.Fields(mcf_strFreigrenze).Value = mpd_curFreigrenze
.Fields(mcf_strGeaendertAm).Value = mpd_dtmGeaendertAm
.Fields(mcf_strGeaendertVon).Value = mpd_strGeaendertVon
.Update

End With
mp_blnDirty = False

End Sub

' ___ Private Methods ___

Private Function iNz(ByVal Value As Variant, ByVal NullValue As Variant) As Variant
If Value = NullValue Then

iNz = Null
Else

iNz = Value
End If

End Function

Implementierung CKunden:
Anforderungen

♦ Kernfeature
• Laden von Tabelle

♦ Komplette Auflistungsklasse
• Add
• Item
• Count
• Remove
• NewEnum – For Each-Tauglichkeit!

♦ Zusätzlich
• Verschiedene Angaben für Items (Index, ID, Objekt)
• Factory-Methode CreateNew
• Leeren mit Clear
• Laden von bestehendem Recordset

Implementierung CKunden:
Features Auflistungsklasse

♦ Interne Collection
Private m_collItems As VBA.Collection

♦ Eigenschaften
Public Property Get Count() As Long

Count = m_collItems.Count

End Property

Public Property Get Item(ByVal Index As Long) As CKundeSimple

Set Item = m_collItems.Item(Index)

End Property

Public Property Get ItemByID(ByVal KundenID As Long) As CKundeSimple

'Attribute ItemByID.VB_UserMemId = 0

Set ItemByID = m_collItems.Item(CStr(KundenID))

End Property

Public Property Get NewEnum() As IUnknown

'Attribute NewEnum.VB_UserMemId = -4

Set NewEnum = m_collItems.[_NewEnum]

End Property

Zugriff über Position

Zugriff über KundenID

Enumerator
(For Each)

Paul Rohorzka Klassen konkret 60

♦ Hinzufügen eines bestehenden Kunden-Objekts
Public Function Add(ByRef Kunde As CKundeSimple) _

As CKundeSimple
If Kunde.ID = 0 Then

m_collItems.Add Kunde, "Ptr" & CStr(ObjPtr(Kunde))
Else

m_collItems.Add Kunde, CStr(Kunde.ID)
End If
Set Add = Kunde

End Function

♦ Erzeugen eines neuen Kunden-Objekts
Public Function CreateNew() As CKundeSimple

Dim kndNew As CKundeSimple

Set kndNew = New CKundeSimple
m_collItems.Add kndNew, "Ptr" & CStr(ObjPtr(kndNew))
Set CreateNew = kndNew
Set kndNew = Nothing

End Function

Keine gültige ID bei
neuen Objekten

Stattdessen: Adresse

Keine gültige ID bei
neuen Objekten

Stattdessen: Adresse

Implementierung CKunden:
Features Auflistungsklasse (cont'd)



2

Paul Rohorzka Klassen konkret 61

Implementierung CKunden:
Features Auflistungsklasse (cont'd)

♦ Entfernen von Elementen
Public Sub Remove(ByVal Index As Long)

m_collItems.Remove Index

End Sub

Public Sub RemoveByID(ByVal KundenID As Long)

m_collItems.Remove CStr(KundenID)

End Sub

Public Sub RemoveByObject(ByRef Kunde As CKundeSimple)

If Kunde.ID = 0 Then

m_collItems.Remove "Ptr" & CStr(ObjPtr(Kunde))

Else

m_collItems.Remove CStr(Kunde.ID)

End If

End Sub

Public Sub Clear()

Set m_collItems = Nothing

Set m_collItems = New VBA.Collection

End Sub

Implementierung CKunden:
Laden der Daten

♦ Laden über neues Recordset
Public Sub Load(Optional ByVal Filter As String, _

Optional ByVal OrderBy As String)
Dim strSQL As String
Dim rst As DAO.Recordset

strSQL = "SELECT * " & _
"FROM " & mc_strTableName

If Len(Filter) > 0 Then
strSQL = strSQL & " WHERE " & Filter

End If
If Len(OrderBy) > 0 Then

strSQL = strSQL & " ORDER BY " & OrderBy
End If

Set rst = CurrentDb().OpenRecordset(strSQL, dbOpenDynaset)

LoadFromRecordset rst

If Not (rst Is Nothing) Then rst.Close
Set rst = Nothing

End Sub

Implementierung CKunden:
Laden der Daten (cont'd)

♦ Laden von geöffnetem Recordset
Public Sub LoadFromRecordset( _

ByRef rst As DAO.Recordset)

Dim kndNew As CKundeSimple

rst.MoveFirst

Do Until rst.EOF

Set kndNew = New CKundeSimple

kndNew.LoadFromRecordset rst

m_collItems.Add kndNew, CStr(kndNew.ID)

rst.MoveNext

Loop

Set kndNew = Nothing

End Sub

Implementierung CKunden:
Erste Tests

♦ Laden von Kunden und Ausgabe
Public Sub TestKundenSimpleForEach()

Dim Kunden As CKundenSimple

Dim Kunde As CKundeSimple

Set Kunden = New CKundenSimple

Kunden.Load "Knd_Nachname LIKE 'A*'"

For Each Kunde In Kunden

Debug.Print Kunde

Next Kunde

Debug.Print "(" & Kunden.Count & " Kunden)"

Set Kunden = Nothing

End Sub

Implementierung CKunden:
Erste Tests (cont'd)

♦ Zugriff auf und Entfernen von Kunden
Public Sub TestKundenSimpleSingle()

Dim Kunden As CKundenSimple

Dim Kunde As CKundeSimple

Set Kunden = New CKundenSimple

Kunden.Load "Knd_Nachname LIKE 'A*'"

Debug.Print Kunden.Count & " Kunden"

Set Kunde = Kunden.Item(Kunden.Count / 2)

Debug.Print "Mittlerer Kunde: " & Kunde

Kunden.RemoveByObject Kunde

Debug.Print Kunden.Count & " Kunden"

Set Kunde = Nothing

Set Kunden = Nothing

End Sub

' This is CkundenSimple
Option Compare Database
Option Explicit

Private Const mc_strTableName As String = "tblKunden"

Private m_collItems As VBA.Collection

' ___ Initialize/Terminate ___

Private Sub Class_Initialize()
Set m_collItems = New VBA.Collection

End Sub

Private Sub Class_Terminate()
Set m_collItems = Nothing

End Sub

' ___ Public Properties ___

'-- Read only
Public Property Get Count() As Long

Count = m_collItems.Count
End Property

'-- Read only
Public Property Get NewEnum() As IUnknown
'Attribute NewEnum.VB_UserMemId = -4

Set NewEnum = m_collItems.[_NewEnum]
End Property

'-- Read only
Public Property Get Item(ByVal Index As Long) As CKundeSimple

Set Item = m_collItems.Item(Index)
End Property

'-- Read only
Public Property Get ItemByID(ByVal KundenID As Long) As CKundeSimple
'Attribute ItemByID.VB_UserMemId = 0

Set ItemByID = m_collItems.Item(CStr(KundenID))
End Property

' ___ Public Methods ___

Public Sub Load(Optional ByVal Filter As String, Optional ByVal OrderBy As String)
Dim strSQL As String
Dim rst As DAO.Recordset

strSQL = "SELECT * " & _
"FROM " & mc_strTableName

If Len(Filter) > 0 Then
strSQL = strSQL & " WHERE " & Filter

End If
If Len(OrderBy) > 0 Then

strSQL = strSQL & " ORDER BY " & OrderBy
End If

Set rst = CurrentDb().OpenRecordset(strSQL, dbOpenDynaset)

LoadFromRecordset rst

If Not (rst Is Nothing) Then rst.Close
Set rst = Nothing

End Sub

Public Sub LoadFromRecordset(ByRef rst As DAO.Recordset)
Dim kndNew As CKundeSimple

rst.MoveFirst
Do Until rst.EOF

Set kndNew = New CKundeSimple
kndNew.LoadFromRecordset rst
m_collItems.Add kndNew, CStr(kndNew.ID)
rst.MoveNext

Loop

Set kndNew = Nothing
End Sub

Public Function Add(ByRef Kunde As CKundeSimple) As CKundeSimple
If Kunde.ID = 0 Then

m_collItems.Add Kunde, "Ptr" & CStr(ObjPtr(Kunde))
Else

m_collItems.Add Kunde, CStr(Kunde.ID)
End If
Set Add = Kunde

End Function

Public Function CreateNew() As CKundeSimple
Dim kndNew As CKundeSimple

Set kndNew = New CKundeSimple
m_collItems.Add kndNew, "Ptr" & CStr(ObjPtr(kndNew))
Set CreateNew = kndNew
Set kndNew = Nothing

End Function

Public Sub Remove(ByVal Index As Long)
m_collItems.Remove Index

End Sub

Public Sub RemoveByID(ByVal KundenID As Long)
m_collItems.Remove CStr(KundenID)

End Sub

Public Sub RemoveByObject(ByRef Kunde As CKundeSimple)
If Kunde.ID = 0 Then

m_collItems.Remove "Ptr" & CStr(ObjPtr(Kunde))
Else

m_collItems.Remove CStr(Kunde.ID)
End If

End Sub

Public Sub Clear()
Set m_collItems = Nothing
Set m_collItems = New VBA.Collection

End Sub

Implementierung CKunden:
Verweis auf Auflistungsobjekt

♦ Problem:
• CKunden muss wissen, wann ein neues Objekt gespeichert wird
• Direkte Referenz CKunde → CKunden ergäbe zirkuläre Referenz

♦ Lösung: „ParentGetter-Pattern“
• Ergänztes Objektmodell:

CKundeCKunde

CKunden CKundenParentGetter

Ereignis

Implementierung CKunden:
„ParentGetter-Pattern“

♦ CKundenParentGetter
Public Event GetParentReference( _

ByRef Kunden As CKunden)

Public Function GetParent() As CKunden

Dim Kunden As CKunden

RaiseEvent GetParentReference(Kunden)

Set GetParent = Kunden

End Function

♦ In CKunden
Private WithEvents ParentGetter As CKundenParentGetter

...

Private Sub ParentGetter_GetParentReference( _
Kunden As CKunden)

Set Kunden = Me

End Sub

EINFÜGEN:
Code CKunde, Code CKunden, CKundenParentGetter

♦ Detaillierten Code siehe Demos auf der AEK-
Teilnehmer-CDs!

IV. Angewandtes Wissen
Anwendungsklassen

CApplication

Spezialisierte Anwendungsklasse

CTempBackends

CApplication
Erweiterung für Access.Application

♦ Bietet immer wieder neu Programmiertes

♦ Unabhängig von der Problemstellung

♦ Eigenschaften
• Title

- Kapselt CurrentDb().Properties("AppTitle")
• IsMDE

- Info, ob Anwendung als MDE läuft
• SharedDatabase

- Hält Referenz auf CurrentDb()

♦ Methoden
• MsgBox

- Zeigt bei fehlendem Title den Anwendungstitel an

C<MyApp>
Spezialisierte Anwendungsklasse

♦ Platz für globale Variablen und Objekte
• Instanz von CApplication
• User-/Rechner-Infos oä.
• Instanz von CTempBackends

♦ Senke für globale Ereignisse
• Optionsänderung
• ShutDown (mit Watchdog-Formular)

♦ Versionsinformationen

♦ Protokollierungsmethode

♦ Abspeichern von Einstellungen
• Spezifischer und flexibler als VBA.SaveSetting (Enum!)

CTempBackends
Temporäre Backends leicht gemacht

♦ Hilfstabellen
• Lokal: starkes Aufblähen des Frontends
• Extern: beinahe keine Vergrößerung des Frontends

♦ Verwaltet beliebig viele temporäre Backends
• Unterscheidung mittels Key

♦ Backends werden automatisch ...
• ... angelegt
• ... gelöscht

♦ Methode ReconnectTable
• Neueinbinden einer Hilfstabelle in einem temporären Backend

EINFÜGEN:
Code CApplication, CAekDemo, CTempBackends

♦ Detaillierten Code siehe Demos auf der AEK-
Teilnehmer-CDs!

Persönliches Fazit

J Programmieren mit Klassen
ist klasse!

K Klassen schreiben
kann mühsam sein.

... aber auch viel Spaß machen!


